
1. Overview

With no changes to the TTCN source code of LTE/SAE test suite ‘iwd-TTCN3-B2015-03_D15wk25’, test case 8.7.1 can be demonstrated to run correctly with LTE-A UE (see section 5) in LTE-FDD mode. Execution log is provided as evidence

3. Verification Test Summary

Test Case:
TC_8_7_1
Test Group:
\LTE_A\
ATS Version:
iwd-TTCN3-B2015-03_D15wk25
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Intel XMM 7360
Verification Status:
PASS

4. Corrections required for test case 8.7.1
Change 1
	Function Name
	f_TC_8_7_1_EUTRA

	Reason for change
	1) EUTRA Cells are not initialised yet. Hence, these parameters needs to be intialised after EUTRA Cells are initialised
2)  As there is a PDP context exists in UTRAN, true value needs to be passed to function    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT
3)  Authentication parameters needs to be sent to UTRAN after TAU has been on EUTRAN

	Summary of change
	1)  Parameters are initialised after invoke of f_EUTRA_Init
2) Value ‘true’ passed to f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT
3) Authetication parameters sent UTRAN using f_EUTRA_SendAuthParameters

	Source of change
	LTE_A\8_7\RRC_ANR.ttcn

	MCC160
	Accepted


Before:
	function f_TC_8_7_1_EUTRA() runs on EUTRA_PTC

  {

    var boolean v_CSInOtherRAT := false;

    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;

    var integer v_T1_RS_EPRE_Cell1 := -60;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    var NAS_PlmnId v_PlmnId := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);

    var CellIdentity v_CellIdentity1 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell1);
    f_EUTRA_Init (c4);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    //Send Cell information to UTRA PTC

    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN,

                                         f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),

                                         f_ConvertDL_BandwidthToInteger(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)),

                                         cs_EUTRA_HOInfo (f_ConvertUL_BandwidthToInteger (f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1)),

                                                          f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                          f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1)),

                                         omit,  //p_CATested

                                         v_PlmnId,

                                         v_CellIdentity1,

                                         v_TAC);

    //Receive new authentication parameters and set initial cell power level according to T0

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    f_EUTRA_SetCellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE);

    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 3" siclog@

    //The SS changes Cell 1 and Cell 5 levels according to the row "T1" in table 8.7.1.3.2-1

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    f_EUTRA_SetCellPower (eutra_Cell1, v_T1_RS_EPRE_Cell1);

    //@siclog "Step 4" siclog@

    //Generic test procedure in TS 36.508 Table 6.4.2.7A is performed on Cell 1.

    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT(eutra_Cell1, v_CSInOtherRAT, false, false, v_IRAT_Coordination_MSG.OctetData, NORMAL);
    //@siclog "Step 5" siclog@

    //Wait for 6 s for UE to receive system information.

    f_Delay(6.0);

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1 and Cell 5 levels according to the row "T2" in table 8.7.1.3.2-1.

    f_EUTRA_SetCellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE);

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    //Steps 7 to 14 done in UTRA RAT so wait for UTRAN side to finish

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    f_EUTRA_TestBody_Set(false);

    //Remove the non-camping cell

    f_EUTRA_ReleaseAllCells( );

  }



After:
	    function f_TC_8_7_1_EUTRA() runs on EUTRA_PTC

  {

    var boolean v_CSInOtherRAT := false;

    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;

    var integer v_T1_RS_EPRE_Cell1 := -60;

    var GutiParameters_Type v_GutiParams;

    var NAS_PlmnId v_PlmnId;

    var TrackingAreaCode v_TAC;

    var CellIdentity v_CellIdentity1;

    f_EUTRA_Init (c4);

    //Create and configure all cells

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell1, -60); 
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    v_PlmnId := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    v_TAC := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);

    v_CellIdentity1 := f_EUTRA_CellInfo_GetCellIdentity (eutra_Cell1);
    //Send Cell information to UTRA PTC

    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN,

                                         f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),

                                         f_ConvertDL_BandwidthToInteger(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)),

                                         cs_EUTRA_HOInfo (f_ConvertUL_BandwidthToInteger (f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1)),

                                                          f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                          f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1)),

                                         omit,  //p_CATested

                                         v_PlmnId,

                                         v_CellIdentity1,

                                         v_TAC);

    //Receive new authentication parameters and set initial cell power level according to T0

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN); //Sent during f_UTRAN_UE_64kPS_Rb_Est

    f_EUTRA_SetCellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE);

    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;

    f_EUTRA_TestBody_Set(true);

    //@siclog "Step 3" siclog@

    //The SS changes Cell 1 and Cell 5 levels according to the row "T1" in table 8.7.1.3.2-1

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    f_EUTRA_SetCellPower (eutra_Cell1, v_T1_RS_EPRE_Cell1);


//tmp NK


//UE has PDP context in UMTS, hence, true needs to be passed

    //@siclog "Step 4" siclog@

    //Generic test procedure in TS 36.508 Table 6.4.2.7A is performed on Cell 1.

    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT(eutra_Cell1, v_CSInOtherRAT, true, false, v_IRAT_Coordination_MSG.OctetData, NORMAL);
    //@siclog "Step 5" siclog@

    //Wait for 6 s for UE to receive system information.

    f_Delay(6.0);

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1 and Cell 5 levels according to the row "T2" in table 8.7.1.3.2-1.

    f_EUTRA_SetCellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE);

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);

    // send Auth parameters

    f_EUTRA_SendAuthParameters(eutra_Cell1, UTRAN);
    //Steps 7 to 14 done in UTRA RAT so wait for UTRAN side to finish

    f_IRAT_WaitForCoOrd_Trigger(UTRAN);

    f_EUTRA_TestBody_Set(false);

    //Remove the non-camping cell

    f_EUTRA_ReleaseAllCells( );

  }



Change 2
	Function Name
	f_TC_8_7_1_UTRAN

	Reason for change
	1) EUTRA PTC does not wait for system information. Hence, it needs to be ignored
2)  Acc. to 36.508, Table 4.4.4.1-1, SIB19 needs to be set Utra priority as 3 instead of 5
3) As EUTRA does not wait for authentication parameters and not required, hence, authentication parameters are not sent. Also, look for change 6
4)  EUTRA expects release of the UE to perfrom the TAU procedure, hence, capability information sent. Also, looks for change 5

	Summary of change
	1) Removed the sending of system information
2) Modified prority to 3 from 5

3) Passed false as input in the second argument

4) Release information sent using f_IRAT_SendCoOrd

	Source of change
	LTE_A\8_7\RRC_ANR_UTRAN.ttcn

	MCC160
	3) Rejected.  EUTRA does wait, and use, auth parameters in    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
1, 2, 4) Accepted


Before:
	function f_TC_8_7_1_UTRAN ( ) runs on UTRAN_PTC

  {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    //Initialise all UTRAN cells according to 34.108

    f_UTRAN_Init(EUTRA_UTRAN);

    // Cell1 started switched on with default power level

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    //Get Cell Parameters for EUTRA cell

    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);

    //Send the cell info to the EUTRA component

    f_UTRAN_Send_IR_SysInfoToEUTRA ( utran_Cell5 );

    //Set sysinfo in SIB19

    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def (5),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                 v_EUTRA_MeasurementBandwidth,

                                                                                                 4)));

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    //Bring UE to initial state

    f_UTRAN_Preamble(utran_Cell5);
    //Interactive or background / UL:64 DL: 64 kbps / PS RAB + UL:3.4 DL:3.4 kbps SRBs for DCCH

    f_UTRAN_UE_64kPS_Rb_Est(utran_Cell5); //@sic R5-134682 sic@

    f_UTRAN_TestBody_Set(true);

    fl_TC_8_7_1_Body (v_EUTRASysInfo);

    f_UTRAN_TestBody_Set(false);

    //Inform EUTRA PTC that TC is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //Release UTRAN Cells

    f_UTRAN_Postamble(utran_Cell5, U2_CONNECTED);

  };


After:
	      function f_TC_8_7_1_UTRAN ( ) runs on UTRAN_PTC

  {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    //Initialise all UTRAN cells according to 34.108

    f_UTRAN_Init(EUTRA_UTRAN);

    // Cell1 started switched on with default power level

    f_UTRAN_SS_CreateCellDCH(utran_Cell5);

    //Get Cell Parameters for EUTRA cell

    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);

        //Send the cell info to the EUTRA component

    //f_UTRAN_Send_IR_SysInfoToEUTRA ( utran_Cell5 );
    //Set sysinfo in SIB19

    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,

                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def (3),

                                                           cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                 v_EUTRA_MeasurementBandwidth,

                                                                                                 4)));

    f_UTRAN_SendDefSysInfo(utran_Cell5);

    //Bring UE to initial state

    f_UTRAN_Preamble(utran_Cell5, false); //tmp NK

    //Interactive or background / UL:64 DL: 64 kbps / PS RAB + UL:3.4 DL:3.4 kbps SRBs for DCCH

    f_UTRAN_UE_64kPS_Rb_Est(utran_Cell5); //@sic R5-134682 sic@

    //This needs to be sent based on the release version it has TAU will be recieved

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get())); 

    f_UTRAN_TestBody_Set(true);

    fl_TC_8_7_1_Body (v_EUTRASysInfo);

    f_UTRAN_TestBody_Set(false);

    //Inform EUTRA PTC that TC is over

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    //Release UTRAN Cells

    f_UTRAN_Postamble(utran_Cell5, U2_CONNECTED);

  };


Change 3
	Function Name
	fl_TC_8_7_1_Body

	Reason for change
	1) Receive the authentication parameters from ETURA after TAU procedure
2) Need to hanld SERVICE REQUEST and Security procedure needs to be established

#draft ProseCR would be raised for next RAN5 meeting

	Summary of change
	1) Received authentication parameters using f_UTRAN_InterRAT_InitialiseAuthParams
2) Handled service requesting and securithy procedure

	Source of change
	LTE_A\8_7\RRC_ANR_UTRAN.ttcn

	MCC160
	Accepted


Before:
	function fl_TC_8_7_1_Body(IRAT_CoOrd_SysInfo_Type p_EUTRASysInfo) runs on UTRAN_PTC

  {

    var B32_Type v_P_tmsi := oct2bit(px_PTMSI_Def);

    var B28_Type v_CellId;

…..

//@siclog "Step 3" siclog@

    //The SS switches its downlink transmission power settings to column "T1" of table 8.7.1.3.2-1

    f_UTRAN_SetCellPower (utran_Cell5, -70, -72);

    //Synchronise with EUTRA PTC

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    // Wait for the trigger from EUTRA steps 4-5 to continue the test case

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //@siclog "Step 6" siclog@

    //The SS switches its downlink transmission power settings to column "T2" of table 8.7.1.3.2-1

    f_UTRAN_SetCellPower (utran_Cell5, -60, -62);

    //@siclog "Step 7" siclog@

    //Generic test procedure in TS 36.508 Table 6.4.2.8 is performed on Cell 5.

    f_UTRAN_508CheckCampOnNewUtranCell(utran_Cell5, cell_Dch);

    //@siclog "Step 8" siclog@

    //Waiting for 10 seconds to allow UE to activate ANR logging in IDLE mode state.

    f_Delay(10.0);  //@sic R5-134682 sic@

    //@siclog "Step 9" siclog@

    //Page UE to trigger RRC Connection establishment

    f_UTRAN_Page (utran_Cell5, terminatingInteractiveCall, v_P_tmsi);

    //@siclog "Step 10-12" siclog@

    //Complete RRC Connection setup and verify if IE "ANR Logging Results Available" is presentt

    f_UTRAN_RRC_ConnEst(utran_Cell5, terminatingInteractiveCall, true);

    //@siclog "Step 13-14" siclog@

    //Send UE INFORMATION REQUEST message and receive UE INFORMATION RESPONSE message, check if IEs were set properly

    v_PLMN_Identity :=f_UTRAN_Nas2Asn_PlmnId(p_EUTRASysInfo.Eutra[0].PLMN);

    v_TrackingAreaCode := p_EUTRASysInfo.Eutra[0].TAC;

    v_EUTRA_Cell_Identity := p_EUTRASysInfo.Eutra[0].CellId;

    v_EARFCN := p_EUTRASysInfo.Eutra[0].Arfcn;

    v_EUTRA_PhysicalCellIdentity := p_EUTRASysInfo.Eutra[0].HO.PhysCellId;

    …..  




After:
	    function fl_TC_8_7_1_Body(IRAT_CoOrd_SysInfo_Type p_EUTRASysInfo) runs on UTRAN_PTC

  {

    var B32_Type v_P_tmsi := oct2bit(px_PTMSI_Def);

    var B28_Type v_CellId;
…..

  //@siclog "Step 3" siclog@

    //The SS switches its downlink transmission power settings to column "T1" of table 8.7.1.3.2-1

    f_UTRAN_SetCellPower (utran_Cell5, -70, -72);
    //Synchronise with EUTRA PTC

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_Trigger);

    // Wait for the trigger from EUTRA steps 4-5 to continue the test case

    f_IRAT_WaitForCoOrd_Trigger(EUTRA);

    //Auth parameters needs to be received from E-UTRAN

    f_UTRAN_InterRAT_InitialiseAuthParams();
    //@siclog "Step 6" siclog@

    //The SS switches its downlink transmission power settings to column "T2" of table 8.7.1.3.2-1

    f_UTRAN_SetCellPower (utran_Cell5, -60, -62);

    //@siclog "Step 7" siclog@

    //Generic test procedure in TS 36.508 Table 6.4.2.8 is performed on Cell 5.

    f_UTRAN_508CheckCampOnNewUtranCell(utran_Cell5, cell_Dch);

    //@siclog "Step 8" siclog@

    //Waiting for 10 seconds to allow UE to activate ANR logging in IDLE mode state.

    f_Delay(10.0);  //@sic R5-134682 sic@

    //@siclog "Step 9" siclog@

    //Page UE to trigger RRC Connection establishment

    f_UTRAN_Page (utran_Cell5, terminatingInteractiveCall, v_P_tmsi);

    //@siclog "Step 10-12" siclog@

    //Complete RRC Connection setup and verify if IE "ANR Logging Results Available" is presentt

    f_UTRAN_RRC_ConnEst(utran_Cell5, terminatingInteractiveCall, true);

    f_UTRAN_GMM_ServReq (ps_domain);

    f_UTRAN_RRC_Security(utran_Cell5,                        // SECURITY MODE COMMAND, SECURITY MODE COMPLETE

                           false,

                           ps_domain);
    //@siclog "Step 13-14" siclog@

    //Send UE INFORMATION REQUEST message and receive UE INFORMATION RESPONSE message, check if IEs were set properly

    v_PLMN_Identity :=f_UTRAN_Nas2Asn_PlmnId(p_EUTRASysInfo.Eutra[0].PLMN);

    v_TrackingAreaCode := p_EUTRASysInfo.Eutra[0].TAC;

    v_EUTRA_Cell_Identity := p_EUTRASysInfo.Eutra[0].CellId;

    v_EARFCN := p_EUTRASysInfo.Eutra[0].Arfcn;

    v_EUTRA_PhysicalCellIdentity := p_EUTRASysInfo.Eutra[0].HO.PhysCellId;
…..


Change 4
	Function Name
	f_UTRAN_RetrieveANRMeasurements_InterRat

	Reason for change
	UE information response would be received on SRB4 instead of SRB2
#draft proseCR would be rasied to clarity the same

	Summary of change
	Used SRB4 instead of SRB2 to received UE information response

	Source of change
	Common\UTRAN\UTRAN_ANR_CommonFunctions.ttcn

	MCC160
	Accepted conditional to prose CR acceptance


Before:
	function f_UTRAN_RetrieveANRMeasurements_InterRat(UTRAN_CellId_Type p_CellId,

                                                    boolean p_IsGSM,

                                                    template (present) LoggedANRReportInfoList p_ExpectedLoggedANRReportInfoList,

                                                    integer p_ExpectedMatches) runs on UTRAN_PTC return boolean

  {

    var B28_Type v_CellId;

    var UTRAN_SecurityInfo_Type v_SecurityInfo;

    var U_RLC_AM_IND v_DataInd;

    var LoggedANRReportInfoList v_ReceivedLoggedANRReportInfoList;

    var integer i, j;

    var integer v_FoundMatches := 0;

    v_CellId := f_UTRAN_GetCellIndentity(p_CellId);

    //Send UE INFORMATION REQUEST message

    v_SecurityInfo := f_UTRAN_Security_Get ();

    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cds_UEInformationRequest_LoggedANRReportRequest(v_SecurityInfo.dl_IntegrityCheckInfo)));

    // Receive UE INFORMATION RESPONSE message and verify if IEs are set correctly

    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB2, cr_UEInformationResponse_LoggedANRReport_Any)) -> value v_DataInd;
    v_ReceivedLoggedANRReportInfoList := v_DataInd.data.aM_message.uL_DCCH_Message.message_.ul_DCCH_MessageType_ext.ueInformationResponse.loggedANRReportInfoList;

    for (i:= 0; i< lengthof(p_ExpectedLoggedANRReportInfoList); i:=i+1){


After:
	        function f_UTRAN_RetrieveANRMeasurements_InterRat(UTRAN_CellId_Type p_CellId,

                                                    boolean p_IsGSM,

                                                    template (present) LoggedANRReportInfoList p_ExpectedLoggedANRReportInfoList,

                                                    integer p_ExpectedMatches) runs on UTRAN_PTC return boolean

  {

    var B28_Type v_CellId;

    var UTRAN_SecurityInfo_Type v_SecurityInfo;

    var U_RLC_AM_IND v_DataInd;

    var LoggedANRReportInfoList v_ReceivedLoggedANRReportInfoList;

    var integer i, j;

    var integer v_FoundMatches := 0;

    v_CellId := f_UTRAN_GetCellIndentity(p_CellId);

    //Send UE INFORMATION REQUEST message

    v_SecurityInfo := f_UTRAN_Security_Get ();

    U_AM.send(cas_RLC_Data_Req_NoCnf(utran_CellDedicated, tsc_RB2, cds_UEInformationRequest_LoggedANRReportRequest(v_SecurityInfo.dl_IntegrityCheckInfo)));

    /
    // Receive UE INFORMATION RESPONSE message and verify if IEs are set correctly

    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB4, cr_UEInformationResponse_LoggedANRReport_Any)) -> value v_DataInd;
    v_ReceivedLoggedANRReportInfoList := v_DataInd.data.aM_message.uL_DCCH_Message.message_.ul_DCCH_MessageType_ext.ueInformationResponse.loggedANRReportInfoList;

    for (i:= 0; i< lengthof(p_ExpectedLoggedANRReportInfoList); i:=i+1){



Change 5
	Function Name
	f_GetTestcaseAttrib_RetrieveEutraCapabilities

	Reason for change
	To UE capabilities needs to be stored to pass to ETURA PTC as described in chagne 2. Hence, Testcase needs to be added

	Summary of change
	Added 8.7.1 to the list

	Source of change
	LTE_A\Common\TestcaseProperties.ttcn

	MCC160
	Accepted


Before:
	function f_GetTestcaseAttrib_RetrieveEutraCapabilities(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_6_2_3_35") { return true; }

      case ("TC_6_2_4_1") { return true; }

      case ("TC_6_2_4_2") { return true; }

      case ("TC_6_2_4_3") { return true; }

      case ("TC_6_2_4_4") { return true; }

      case ("TC_6_2_4_5") { return true; }

      case ("TC_6_2_4_6") { return true; }

      case ("TC_6_2_4_7") { return true; }

    }

    return false;

  }


After:
	    function f_GetTestcaseAttrib_RetrieveEutraCapabilities(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_6_2_3_35") { return true; }

      case ("TC_6_2_4_1") { return true; }

      case ("TC_6_2_4_2") { return true; }

      case ("TC_6_2_4_3") { return true; }

      case ("TC_6_2_4_4") { return true; }

      case ("TC_6_2_4_5") { return true; }

      case ("TC_6_2_4_6") { return true; }

      case ("TC_6_2_4_7") { return true; }

      case ("TC_8_7_1") { return true; } 

    }

    return false;

  }


Change 6
	Function Name
	f_UTRAN_Preamble

	Reason for change
	As EUTRA does not wait for authentication parameters and not required, hence, authentication parameters are not sent

	Summary of change
	Authentication parameters sent based on ‘AuthparamsTobeSent’ flag

	Source of change
	Common\UTRAN\UTRAN_CommonProcedures.ttcn

	MCC160
	Rejected.  Not required.


Before:
	function f_UTRAN_Preamble(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC

  { /* @sic R5s130347: local GuardTimer sic@ */

    var UTRAN_SecurityInfo_Type v_SecurityInfo;

    var Common_AuthenticationParams_Type v_AuthCS;

    timer t_GuardTimer := tsc_GuardTimePreamble;

    t_GuardTimer.start;

    //  Switch cell power on

    f_UTRAN_SetCellPower(p_CellId, tsc_AttenuationServingCell);

    //  Switch on UE

    f_UT_SwitchOnUE(UT, false);

    //  Establish RRC Connection, perform Authentication, Security Mode procedure,

    //  Attach procedure and release RRC Connection

    f_UTRAN_IdleUpdated(p_CellId);

    // Now get the security parameters that have been updated

    v_SecurityInfo := f_UTRAN_Security_Get();

    // AuthCS not the same structure as AuthPS

    v_AuthCS := v_SecurityInfo.authKeys.AuthPS;

    v_AuthCS.CK := v_SecurityInfo.authKeys.AuthCS.CKey;

    v_AuthCS.IK := v_SecurityInfo.authKeys.AuthCS.IKey;

    v_AuthCS.KeySeq := v_SecurityInfo.authKeys.AuthCS.CKeySeq;

    EUTRA.send(cms_IRAT_AuthInfo(v_SecurityInfo.authKeys.AuthPS,

                                 v_AuthCS,

                                 omit,  // may need to send this eventually

                                 v_SecurityInfo.authKeys.CSinOtherRAT));

    t_GuardTimer.stop;

  }




After:
	  function f_UTRAN_Preamble(UTRAN_CellId_Type p_CellId, boolean AuthParamsTobeSent := true) runs on UTRAN_PTC

  { /* @sic R5s130347: local GuardTimer sic@ */

    var UTRAN_SecurityInfo_Type v_SecurityInfo;

    var Common_AuthenticationParams_Type v_AuthCS;

    timer t_GuardTimer := tsc_GuardTimePreamble;

    t_GuardTimer.start;

    //  Switch cell power on

    f_UTRAN_SetCellPower(p_CellId, tsc_AttenuationServingCell);

    //  Switch on UE

    f_UT_SwitchOnUE(UT, false);

    //  Establish RRC Connection, perform Authentication, Security Mode procedure,

    //  Attach procedure and release RRC Connection

    f_UTRAN_IdleUpdated(p_CellId);

    // Now get the security parameters that have been updated

    v_SecurityInfo := f_UTRAN_Security_Get();

    // AuthCS not the same structure as AuthPS

    v_AuthCS := v_SecurityInfo.authKeys.AuthPS;

    v_AuthCS.CK := v_SecurityInfo.authKeys.AuthCS.CKey;

    v_AuthCS.IK := v_SecurityInfo.authKeys.AuthCS.IKey;

    v_AuthCS.KeySeq := v_SecurityInfo.authKeys.AuthCS.CKeySeq;

    if(AuthParamsTobeSent) {
    EUTRA.send(cms_IRAT_AuthInfo(v_SecurityInfo.authKeys.AuthPS,

                                 v_AuthCS,

                                 omit,  // may need to send this eventually

                                 v_SecurityInfo.authKeys.CSinOtherRAT));

    }
    t_GuardTimer.stop;

  }



Change 7
	Function Name
	f_UTRAN_RRC_ReceiveConnSetupCmpl

	Reason for change
	‘true_’ used then compiler will not know from where to pick the correct value. True_ can be ‘0’ or ‘1’ based on the definition,
altE-bitInterpretation ENUMERATED {false (0), true (1)}

To make it more explicit to the compiler, new variable defined to point to the correct definition

	Summary of change
	Defined new variable v_RRCConnectionSetupComplete_va40ext_IEs_Exp whjich will be used during the match operation

	Source of change
	Common\UTRAN\UTRAN_CommonFunctions.ttcn

	MCC160
	Implemented alternative solution: swapped parameters in match operation:
match(v_RRCConnectionSetupComplete_va40ext_IEs.loggedANRResultsAvailable, true_);


Before:
	function f_UTRAN_RRC_ReceiveConnSetupCmpl(UTRAN_CellId_Type p_CellId,

                                            template (omit) boolean p_LogResultsAvail_ANR := omit)  //@sic R5-134682 sic@

    runs on UTRAN_PTC

  {

    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);

    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();

    var RRCConnectionSetupComplete v_RRCConnectionSetupComplete;

    var U_RLC_AM_IND v_RLC_AM_DATA_IND;

    var STARTList v_STARTList;

    var boolean v_ConfigIsDCH;

    var UE_CapabilityContainer_IEs v_UE_CapabilityContainer;

    var bitstring v_UeCapabilityContainerBitstring;

    var boolean v_LogMeasAvailReceived_ANR, v_LogMeasAvailExpected_ANR;

    var RRCConnectionSetupComplete_va40ext_IEs v_RRCConnectionSetupComplete_va40ext_IEs;

    select (v_CellConfig) {
….

    // Check if ANR measurement log available flag

    if (isvalue(p_LogResultsAvail_ANR)) {

      v_LogMeasAvailExpected_ANR := valueof(p_LogResultsAvail_ANR);

      v_LogMeasAvailReceived_ANR := false;

      if (ispresent(v_RRCConnectionSetupComplete.v370NonCriticalExtensions.v380NonCriticalExtensions.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v590NonCriticalExtensions.v5c0NonCriticalExtensions.v690NonCriticalExtensions.v770NonCriticalExtensions.va40NonCriticalExtensions.rrcConectionSetupComplete_va40ext)) {

        v_RRCConnectionSetupComplete_va40ext_IEs := v_RRCConnectionSetupComplete.v370NonCriticalExtensions.v380NonCriticalExtensions.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v590NonCriticalExtensions.v5c0NonCriticalExtensions.v690NonCriticalExtensions.v770NonCriticalExtensions.va40NonCriticalExtensions.rrcConectionSetupComplete_va40ext;

        v_LogMeasAvailReceived_ANR := ispresent(v_RRCConnectionSetupComplete_va40ext_IEs.loggedANRResultsAvailable) and

          match(true_, v_RRCConnectionSetupComplete_va40ext_IEs.loggedANRResultsAvailable);  // NOTE: the "match" would not be needed as long as "true" is the only value

      }

      if (v_LogMeasAvailExpected_ANR == v_LogMeasAvailReceived_ANR) {   // Verify that the value of IE Logged ANR  Results Available is the same as expected

        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Correct IE  Logged ANR Results Available");

      } else {

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Incorrect IE Logged ANR Results Available");

      }

    }

  }




After:
	  function f_UTRAN_RRC_ReceiveConnSetupCmpl(UTRAN_CellId_Type p_CellId,

                                            template (omit) boolean p_LogResultsAvail_ANR := omit)  //@sic R5-134682 sic@

    runs on UTRAN_PTC

  {

    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);

    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();

    var RRCConnectionSetupComplete v_RRCConnectionSetupComplete;

    var U_RLC_AM_IND v_RLC_AM_DATA_IND;

    var STARTList v_STARTList;

    var boolean v_ConfigIsDCH;

    var UE_CapabilityContainer_IEs v_UE_CapabilityContainer;

    var bitstring v_UeCapabilityContainerBitstring;

    var boolean v_LogMeasAvailReceived_ANR, v_LogMeasAvailExpected_ANR;

    var RRCConnectionSetupComplete_va40ext_IEs v_RRCConnectionSetupComplete_va40ext_IEs;

    var RRCConnectionSetupComplete_va40ext_IEs v_RRCConnectionSetupComplete_va40ext_IEs_Exp;
    select (v_CellConfig) {    
…..

    }

    // Check if ANR measurement log available flag

    if (isvalue(p_LogResultsAvail_ANR)) {

      v_LogMeasAvailExpected_ANR := valueof(p_LogResultsAvail_ANR);

      v_RRCConnectionSetupComplete_va40ext_IEs_Exp.loggedANRResultsAvailable := true_;
      v_LogMeasAvailReceived_ANR := false;

      if (ispresent(v_RRCConnectionSetupComplete.v370NonCriticalExtensions.v380NonCriticalExtensions.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v590NonCriticalExtensions.v5c0NonCriticalExtensions.v690NonCriticalExtensions.v770NonCriticalExtensions.va40NonCriticalExtensions.rrcConectionSetupComplete_va40ext)) {

        v_RRCConnectionSetupComplete_va40ext_IEs := v_RRCConnectionSetupComplete.v370NonCriticalExtensions.v380NonCriticalExtensions.v3a0NonCriticalExtensions.laterNonCriticalExtensions.v3g0NonCriticalExtensions.v4b0NonCriticalExtensions.v590NonCriticalExtensions.v5c0NonCriticalExtensions.v690NonCriticalExtensions.v770NonCriticalExtensions.va40NonCriticalExtensions.rrcConectionSetupComplete_va40ext;

        v_LogMeasAvailReceived_ANR := ispresent(v_RRCConnectionSetupComplete_va40ext_IEs.loggedANRResultsAvailable) and match(v_RRCConnectionSetupComplete_va40ext_IEs_Exp.loggedANRResultsAvailable, v_RRCConnectionSetupComplete_va40ext_IEs.loggedANRResultsAvailable);  // NOTE: the "match" would not be needed as long as "true" is the only value

      }

      if (v_LogMeasAvailExpected_ANR == v_LogMeasAvailReceived_ANR) {   // Verify that the value of IE Logged ANR  Results Available is the same as expected

        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Correct IE  Logged ANR Results Available");

      } else {

        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Incorrect IE Logged ANR Results Available");

      }

    }

  }


5. Execution Log Files

5.1 Intel XMM 7360 
The Intel XMM7360 UE passed this LTE-FDD test case on Anritsu Protocol Conformance Test System ME7832L in FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
References

	[1]
	R5s150754:    Supporting information for agreement of TC 8.7.1 in LTE-FDD mode.
                          This archive comprises:

                        -  html and text format execution log files 



